/*

Hifiduino v 0.5

April 25, 2008

In this release:

- LCD Showing Filter selection and Volume level.

- Digital Filter Selection: Filter 1 to Filter 5.

- Volume Control. Now -99 db to 0db in 1db increment. Using buttons.

- Enabled anti-clipping and volume ramp-up/down when changing volume.

  Anti clipping reduces volume by -2db preventing clipping distortion because

  modern recordings are done very close to 0db.

- Removed the delay() in the write to register routine. Increase volume

  changing speed 2X now is -99 db to 0 db in 10 seconds. Used to be 20 sec.

- Volume "dimmer". Volume dims to -80 dB at the push of a button.

  Pushing any of the volume buttons will "undim".

- Changed de default delay in the LCD library to (0,0) See below note.

- Because of the above, the loop is very fast so that when pressing the

  buttons, the loop will go through several times, so I put a delay.

- Support for rotary encoders (at least mechanical types). Uses a single

  interrupt line to detect motion of the rotary encoder. The chaning of the

  volume is done outside of the interrupt service routine, because 1- I2C doesn't

  work inside the interrupt routine and 2- interrupt routines should be brief.

       --NEW---NEW---NEW---

  Remote code adapted from Arduino forums. For Sony remote, interrupt-based

  and non-blocking

  Pin 2 and pin 3 of the IR receiver are in GND and +5 respectively

  Pin 1 of the IR receiver is connected to pin 3 of the Arduino. Pin 3 is pulled high by s/w

   (I changed from pin 2 in the forum code to pin 3 as I'm using pin 2 for the rotary encoder)

  A quick summary of the Sony TV protocol:

  -The whole set is 13 pulses long

  -The pulses are only measured while low (I'm sure there's a cool word for that but I don't know it)

  -The first pulse, the start pulse, lasts about 2.4 ms (2400 microseconds)

  -The following pulses are 1.2 ms (1200 microseconds) for a 1, 0.6 ms (600 microseconds) for a 0

  -Each data pulse is about 0.3 ms (300 microseconds) apart

  -Every set of 13 pulses is about 20-30 ms apart from each other, repeating when a button is held down

  The code below will watch for a 13 pulse set and returns the set's integer equivalent,

  different for every button on the remote. If you want to try your own TV code you can always set

  the debug flag to 1 and see all of the pretty microsecond measurements.

*/

/*

Additional notes:

- The default delay for the web4robot LCD is (50,4)

  It can be changed with setDelay(Cmd,Char).

  In fact setDelay(0,0) works with the web4robot display

  at least in this code, making it fairly fast

*/

#include <Wire.h>

#include <LCDi2cW.h>

// Define register values, etc to facilitate programming

#define REG9 0x12     // Register 9 address (PCM vs DSD)

#define RESET 0x00    // Write anything to reset the DAC

#define REG7 0x0E     // Register 7 address (PCM vs DSD)

#define PCM 0x40      // Value for reg7 for PCM, High Sample rate for filter

#define REG6 0x0C     // Register 6 address (filters)

#define FILTER1 0x00  // PCM Filter response 1

#define FILTER2 0x01  // PCM Filter response 2

#define FILTER3 0x02  // PCM Filter response 3

#define FILTER4 0x03  // PCM Filter response 4

#define FILTER5 0x04  // PCM Filter response 5

#define REG0 0x00     // Register 0 address (Volume Adjustment)

#define REG1 0x02     // Register 1 address (Volume Adjustment)

#define REG2 0x04     // Register 2 address (Volume Adjustment)

#define REG3 0x06     // Register 3 address (Volume Adjustment)

#define REG4 0x08     // Register 4 address (Volume Control)

#define VOLCNTR 0x07  // VolRight=VolLeft, AntiClip, Vol Ramping is on

#define DEFAULTVOL 0xF0 //-60 dB this is 40x4=160

#define MINVOL 0x18C    //-99dB this is 90X4=396. -Dac adjustment is .25 db

#define MAXVOL 0x00     //-0 dB

#define DIMVOL 0x140    //-80dB The volume level when dimming the volume

#define FILTERPIN 6     // Button to select Filter

#define VOLDIMPIN 5     // Button for volume dimming feature

#define VOLUPPIN 4      // Button to increase  volume

#define VOLDOWNPIN 2    // Button to decrease volume

#define IRPIN 3         // The pin for the remote sensor

#define KEY1 128           // The value when pressing the 1 key in remote

#define KEY2 129           // The value when pressing the 2 key in remote

#define KEY3 130           // The value when pressing the 3 key in remote

#define KEY4 131           // The value when pressing the 4 key in remote

#define KEY5 132           // The value when pressing the 5 key in remote

#define KEYVOLUP 147       // The value when pressing volume up key in remote

#define KEYVOLDOWN 146     // The value when pressing volume down key in remote

#define IRFILSELDELAY 300  // Delay when selecting filter with remote to preven multiple selection

// The write to WM8741 DAC routine

void opuswritereg(uint8_t regaddr,uint8_t regval)

{

  Wire.beginTransmission(0x1B);

  Wire.send(regaddr);

  Wire.send(regval);

  Wire.endTransmission();

}

// Interrupt service routine for rotary encoder. Only determines direction

// of motion and sets flags

volatile byte volUp=0;  // flags for the interrupt routine

volatile byte volDown=0;

void decoder()

{

  if (digitalRead(2) == digitalRead(4))

  {

    volUp = 1;  //if on interrupt the encoder channels are the same, direction is clockwise

  }

  else

  {

    volDown = 1;  //if they are not the same, direction is ccw

  }

}

// interrupt service routine for remote

volatile byte remoteOn = 0;  // 1 means remote has been pressed

void remoting()

{

  remoteOn=1;

}

LCDi2cW lcd = LCDi2cW(4,20,0x4C,0); // Used for the web4robot LCD

// Declaring some variables

int currVol=DEFAULTVOL; // this needs to be at least 10 bits

byte filter=0;          // variable for current filter

byte reg0val=0;         // variable for volume lower bits

byte reg1val=0;         // variable for volume upper bits

byte volDimState=0;     // State for volume dimmer. =1 is dimmed

int startBit = 2000;
// Start bit threshold (Microseconds)

int IRbin1 = 1000;
// Binary 1 threshold (Microseconds)

int IRbin0 = 400;
// Binary 0 threshold (Microseconds)

int longpulse=1;        // If longpulse==0 it means pulseIn() timed out

int irCode=0;           // The code returned by the remote IR pulses

void setup() { 

  lcd.init();     // Initializing the LCD, clears the display

  lcd.setDelay(0,0); // Change de delay in the LCD library 

  Wire.begin();   // Joining the I2C bus as master

  pinMode(FILTERPIN, INPUT);      // Button to select Input fiter

  digitalWrite(FILTERPIN, HIGH);  // Enable pull-up resistor

  pinMode(VOLUPPIN, INPUT);       // Button or Encoder pin for volume up

  digitalWrite(VOLUPPIN, HIGH);   // Enable pull-up resistor

  pinMode(VOLDOWNPIN, INPUT);     // Button or Encoder pin for volume down

  digitalWrite(VOLDOWNPIN, HIGH); // Enable pull-down resistor

  pinMode(VOLDIMPIN, INPUT);      // Button for volume dimmer

  digitalWrite(VOLDIMPIN, HIGH);  // Enable pull-up resistor

  pinMode(IRPIN, INPUT);          // IR Receiver

  digitalWrite(IRPIN, HIGH);      // Enable high as specified by datasheet

  opuswritereg(REG9,RESET);   // Resetting the DAC just in case

  // The following 4 lines is to set up the default volume

  // To set up the volume you need to write to at least 2 registers

  reg0val=(DEFAULTVOL & 0x1F);     // The lower 5 bits goes to reg 0

  opuswritereg(REG0,reg0val);      // Writing to reg 0

  reg1val=((DEFAULTVOL>>5)| 0x20); // The upper 5 bits goes to reg 1

  opuswritereg(REG1,reg1val);      // Writing to reg 1

  opuswritereg(REG4,VOLCNTR);  // Setting up volume control

  opuswritereg(REG6,FILTER1);  // I'm setting the default filter at power on

  opuswritereg(REG7,PCM);      // Setting up for PCM and high sampling rate because I have 

                               // a reclocker and the output is 192K to the DAC

  lcd.print("OPUS Setup Done"); // Just to see we are done with DAC setup

  delay(2000);                  // Let message show for 2 sec.

  lcd.clear();                  // Clears the LCD

  delay(100);

  lcd.print("Filter: RESPONSE #1");  // print default filter

  lcd.setCursor(2,0);

  lcd.print("Volume: -");            // Show startup volume

  lcd.print(currVol/4);

  lcd.print(" dB");

  lcd.setCursor(1,0);

  lcd.print("S Rate: 44.1KHz");

  lcd.setCursor(3,0);

  lcd.print("Source: PCM");

  attachInterrupt(0, decoder, CHANGE);  // This is all that is needed to enable the interrupt

  attachInterrupt(1, remoting, RISING); // ISR for remote IR sensor

}

void loop()

{

  // The following code will select the next filter if the button is pushed

  // and it is repeated for the remote because I'm lazy :-)

  if (digitalRead(FILTERPIN)==0)  // When the filter button is pushed

  {  

    filter++;

    filter=filter%5;  

    lcd.setCursor(0,8);

    switch(filter){

      case 0:

      opuswritereg(REG6,FILTER1);

      lcd.print("RESPONSE #1");

      break;

      case 1:

      opuswritereg(REG6,FILTER2);

      lcd.print("RESPONSE #2");

      break;

      case 2:

      opuswritereg(REG6,FILTER3);

      lcd.print("RESPONSE #3");

      break;

      case 3:

      opuswritereg(REG6,FILTER4);

      lcd.print("RESPONSE #4");

      break;

      case 4:

      opuswritereg(REG6,FILTER5);

      lcd.print("RESPONSE #5");

      break;

    }

    delay(300);  // Used to avoid reading the button more than one during the click

  }

  // The following is to adjust the volume down (larger number)

  //while(digitalRead(VOLDOWNPIN)==0)  //for button

  while(volUp==1)  // While there is CW motion in the rotary encoder

  {

    volUp=0;  // Reset the flag

    if (currVol<MINVOL)   // Check if already at min numerical Volume

    {

      currVol=currVol+4;                    // Increase 1 dB

      reg0val=(((byte)currVol) & 0x1F);     // Calculate value for reg0

      opuswritereg(REG0,reg0val);           // Write to reg0

      reg1val=(((byte)(currVol>>5))| 0x20); // Calculate value for reg1

      opuswritereg(REG1,reg1val);           // Write to reg1

      lcd.setCursor(2,9);

      lcd.print(currVol/4);              // Update the LCD

      if (volDimState==1)                // If in dim state, undim

      {

        lcd.print(" dB      ");

        volDimState=0;

      }

    }

    //delay(120); // Used to avoid reading the button more than one during the click

                  // but not needed for rotary encoder

  }

  // The following is to adjust the volume up (smaller numbers)

  //while(digitalRead(VOLUPPIN)==0)  //This is used for button switch

  while(volDown==1)  // While there is ccw motion in rotary encoder

  {

    volDown=0;  // clear the flag

    if (currVol>MAXVOL)   // Check if already at max Volume

    {

      currVol=currVol-4;                // Decrease 1 dB

      reg0val=(((byte)currVol) & 0x1F); // Calculate value for reg0

      opuswritereg(REG0,reg0val);       // Write to reg0

      reg1val=((currVol>>5)| 0x20);     // Calculate value for reg1

      opuswritereg(REG1,reg1val);       // Write to reg1

      if (currVol==36)                  // Detect from 2 digit to 1 digit

      {                                 // and clean the zero that is left in the screen

        lcd.setCursor(2,9);

        lcd.print("  ");

      }

      lcd.setCursor(2,9);

      lcd.print(currVol/4);             // Update the LCD

      if (volDimState==1)               // If in dim state, undim

      {

        lcd.print(" dB      ");

        volDimState=0;

      }

    }

    //delay(120); // Used to avoid reading the button more than one during the click

                  // but not needed for rotary encoder

  }

  // The following is for the volume dimming feature

  if (digitalRead(VOLDIMPIN)==0) // When pressing the volume dim button

  {

    if (volDimState==0)  //if not dimmed, then dim the volume to min vol

    {

      lcd.setCursor(2,9);

      reg0val=(((byte)DIMVOL) & 0x1F);

      opuswritereg(REG0,reg0val);

      reg1val=(((byte)(DIMVOL>>5))| 0x20);

      opuswritereg(REG1,reg1val);

      lcd.print(DIMVOL/4);

      lcd.print(" dB <DIM>");

      volDimState=1;     // set state to dimmed

    }

    else                 // if dimmed, then return to current volume

    {

      lcd.setCursor(2,9);

      reg0val=(((byte)currVol) & 0x1F);

      opuswritereg(REG0,reg0val);

      reg1val=(((byte)(currVol>>5))| 0x20); // Calculate value for reg1

      opuswritereg(REG1,reg1val);

      lcd.print(currVol/4);

      lcd.print(" dB      ");

      volDimState=0;    // set state to undimmed.

    }

    delay(300); // Used to avoid reading the button more than one during the click

  }

  // The following code is for the remote.

  while (remoteOn==1)

  {

    irCode = getIRKey();

    //Fetch the key

    switch(irCode){

      // The following for filter selection

      case KEY1:

      opuswritereg(REG6,FILTER1);

      lcd.setCursor(0,18);

      lcd.print("1");

      delay(IRFILSELDELAY);

      break;

      case KEY2:

      opuswritereg(REG6,FILTER2);

      lcd.setCursor(0,18);

      lcd.print("2");

      delay(IRFILSELDELAY);

      break;

      case KEY3:

      opuswritereg(REG6,FILTER3);

      lcd.setCursor(0,18);

      lcd.print("3");

      delay(IRFILSELDELAY);

      break;

      case KEY4:

      opuswritereg(REG6,FILTER4);

      lcd.setCursor(0,18);

      lcd.print("4");

      delay(IRFILSELDELAY);

      break;

      case KEY5:

      opuswritereg(REG6,FILTER5);

      lcd.setCursor(0,18);

      lcd.print("5");

      delay(IRFILSELDELAY);

      break;

      // The following for volume

      case KEYVOLUP:

      if (currVol<MINVOL)   // Check if already at min numerical Volume

      {

        currVol=currVol+4;                    // Increase 1 dB

        reg0val=(((byte)currVol) & 0x1F);     // Calculate value for reg0

        opuswritereg(REG0,reg0val);           // Write to reg0

        reg1val=(((byte)(currVol>>5))| 0x20); // Calculate value for reg1

        opuswritereg(REG1,reg1val);           // Write to reg1

        lcd.setCursor(2,9);

        lcd.print(currVol/4);              // Update the LCD

        if (volDimState==1)                // If in dim state, undim

        {

          lcd.print(" dB      ");

          volDimState=0;

        }

      }

      break;

      case KEYVOLDOWN:

      if (currVol>MAXVOL)   // Check if already at max Volume

      {

        currVol=currVol-4;                // Decrease 1 dB

        reg0val=(((byte)currVol) & 0x1F); // Calculate value for reg0

        opuswritereg(REG0,reg0val);       // Write to reg0

        reg1val=((currVol>>5)| 0x20);     // Calculate value for reg1

        opuswritereg(REG1,reg1val);       // Write to reg1

        if (currVol==36)                  // Detect from 2 digit to 1 digit

        {                                 // and clean the zero that is left in the screen

          lcd.setCursor(2,9);

          lcd.print("  ");

        }

        lcd.setCursor(2,9);

        lcd.print(currVol/4);             // Update the LCD

        if (volDimState==1)               // If in dim state, undim

        {

          lcd.print(" dB      ");

          volDimState=0;

        }

      }

      break;

    }

    remoteOn=0;  //reset flag

    longpulse=1; //reset longpulse

  }

}  

// This is the main decoding code for the remote.

int getIRKey() {

  int data[12];

  while((longpulse=pulseIn(IRPIN, LOW, 50000)) < 2200)  //waiting for the long pulse

  {

    if(longpulse==0) // if timed out

    {

      break;

    }

  }

  data[0] = pulseIn(IRPIN, LOW, 2000);      //Start measuring bits, I only want low pulses

  data[1] = pulseIn(IRPIN, LOW, 2000);

  data[2] = pulseIn(IRPIN, LOW, 2000);

  data[3] = pulseIn(IRPIN, LOW, 2000);

  data[4] = pulseIn(IRPIN, LOW, 2000);

  data[5] = pulseIn(IRPIN, LOW, 2000);

  data[6] = pulseIn(IRPIN, LOW, 2000);

  data[7] = pulseIn(IRPIN, LOW, 2000);

  data[8] = pulseIn(IRPIN, LOW, 2000);

  data[9] = pulseIn(IRPIN, LOW, 2000);

  data[10] = pulseIn(IRPIN, LOW, 2000);

  data[11] = pulseIn(IRPIN, LOW, 2000);

  // delay(50); // to slow down the loop if needed

  for(int i=0;i<11;i++)  //Parse the pulses

  { 

    if(data[i] > IRbin1) {
  //is it a 1?


data[i] = 1;

    }  else {


 if(data[i] > IRbin0) {
  //is it a 0?


   data[i] = 0;


 } else {


   data[i] = 2;

  //Flag the data as invalid; I don't know what it is!


   }

       }

  }

  for(int i=0;i<11;i++) {

  //Pre-check data for errors

    if(data[i] > 1) {


return -1;


     //Return -1 on invalid data

    }

  }

  int result = 0;

  int seed = 1;

  for(int i=0;i<11;i++) {

  //Convert bits to integer

    if(data[i] == 1) {


result += seed;

    }

    seed = seed * 2;

  }

  return result;


     //Return key number 

}

